Double Hopf in 1:2 resonance

Modification of "Perturbation Methods with Mathematica", by Ali H.Nayfeh and Char - Ming Chin; email: anayfeh@vt-.edu - cchin@vt.edu

Double Rayleigh-Duffing oscillator in nearly 1:2 resonance. MSM.

$$\ddot{q}_1 - \mu\,\dot{q}_1 + \omega_1^2\,q_1 - b_0\,(\dot{q}_2 - \dot{q}_1)^2 + b_1\,\dot{q}_1^3 + c\,q_1^3 = 0$$
$$\ddot{q}_2 - \nu\,\dot{q}_2 + \omega_2^2\,q_2 + b_0\,(\dot{q}_2 - \dot{q}_1)^2 + b_2\,\dot{q}_2^3 + c\,q_2^3 = 0$$

$$\omega_2 \simeq 2\,\omega_1$$

Time scales and definitions

Utility (don't modify)

```
Off[General::spell1]
```

```
<< Notation`
```

Time scales

```
Symbolize[T₀]; Symbolize[T₁]; Symbolize[T₂];
```

```
timeScales = {T₀, T₁, T₂};
```

Differentiation (don't modify)

```
dt[1][expr_] := Sum[ϵ^i D[expr, timeScales[[i + 1]]], {i, 0, maxOrder}];
dt[2][expr_] := (dt[1][dt[1][expr]] // Expand) /. ϵ^(i_ /; i>maxOrder) -> 0;
```

Some rules (don' t modify)

```
conjugateRule = {A -> Ā, Ā -> A, Γ -> Γ̄, Γ̄ -> Γ, Complex[0, n_] -> Complex[0, -n]};
```

```
displayRule = {q_i_,j_^(a__)[__] :> Row[{Times @@ MapIndexed[D^#1_#2[[1]]-1 &, {a}], q_i,j}],
   A_i_^(a__)[__] :> Row[{Times @@ MapIndexed[D^#1_#2[[1]] &, {a}], A_i}],
   q_i_,j_[__] :> q_i,j, A_i_[__] :> A_i};
```

Uncomment the following if you want to display $\{x_i, y_i\}$ instead of $\{q_{1,i}, q_{2,i}\}$

```
displayRule = {q_1,j_^(a__)[__] :> Row[{Times @@ MapIndexed[d^#1_#2[[1]]-1 &, {a}], x_j}],
   q_2,j_^(a__)[__] :> Row[{Times @@ MapIndexed[d^#1_#2[[1]]-1 &, {a}], y_j}],
   A_i_^(a__)[__] :> Row[{Times @@ MapIndexed[d^#1_#2[[1]] &, {a}], A_i}],
   q_1,j_[__] :> x_j, q_2,j_[__] :> y_j, A_i_[__] :> A_i};
```

Equations of motions and resonance conditions

Equations of motion

```
EOM = { q1''[t] - μ q1'[t] + ω1^2 q1[t] + c q1[t]^3 + b1 q1'[t]^3 - b0 (q2'[t] - q1'[t])^2 == 0,
   q2''[t] - ν q1'[t] + ω̄2^2 q2[t] + c q2[t]^3 + b2 q2'[t]^3 + b0 (q2'[t] - q1'[t])^2 ==
    0}; EOM // TableForm
```

$$\omega_1^2\, q_1[t] + c\, q_1[t]^3 - \mu\, (q_1)'[t] + b_1\, (q_1)'[t]^3 - b_0\, (-(q_1)'[t] + (q_2)'[t])^2 + (q_1)''[t] = 0$$
$$\overline{\omega}_2^2\, q_2[t] + c\, q_2[t]^3 - \nu\, (q_1)'[t] + b_2\, (q_2)'[t]^3 + b_0\, (-(q_1)'[t] + (q_2)'[t])^2 + (q_2)''[t] = 0$$

Ordering of the dampings

```
smorzrule = {ν -> ϵ ν, μ -> ϵ μ};
```

Scaling of the variables

```
scaling = {q1[t] -> ϵ q1[t], q2[t] -> ϵ q2[t], q1'[t] -> ϵ q1'[t],
   q2'[t] -> ϵ q2'[t], q1''[t] -> ϵ q1''[t], q2''[t] -> ϵ q2''[t]}
```

$\{q_1[t] \to \epsilon\, q_1[t],\ q_2[t] \to \epsilon\, q_2[t],\ (q_1)'[t] \to \epsilon\, (q_1)'[t],$
$(q_2)'[t] \to \epsilon\, (q_2)'[t],\ (q_1)''[t] \to \epsilon\, (q_1)''[t],\ (q_2)''[t] \to \epsilon\, (q_2)''[t]\}$

Definition of $\overline{\omega}_2$ (introduction of the detuning)

```
ombrule = {ω̄2 -> ω2 + ϵ σ}
```

$\{\overline{\omega}_2 \to \epsilon\, \sigma + \omega_2\}$

Definition of the expansion of q_i

```
solRule = q_i_ -> (Sum[ϵ^(j-1) q_i,j-1[#1, #2, #3], {j, 3}] &);
```

Some rules (don' t modify)

```
multiScales = {q_i_[t] → q_i @@ timeScales,
   Derivative[n_][q_][t] -> dt[n][q @@ timeScales], t -> T_0};
```

This is the result of "multiScales" and "solRule":

```
q_1[t] /. multiScales /. solRule /. displayRule
```

$$x_0 + \epsilon x_1 + \epsilon^2 x_2$$

Max order of the procedure

```
maxOrder = 2;
```

Modification of the equations of motion: substitution of the rules.

```
EOMa = (EOM /. scaling /. multiScales /. smorzrule /. ombrule /. solRule // TrigToExp //
      ExpandAll) /. ε^n_/;n>3 -> 0;
```

Separation of the coefficients of the powers of ϵ

```
eqEps = Rest[Thread[CoefficientList[Subtract @@ #, ε] == 0]] &/@EOMa // Transpose;
```

Definition of the equations at orders of ϵ and representation

```
eqOrder[i_] := (#[[1]] & /@ eqEps[[1]] /. q_k_,0 → q_k,i-1) ==
   (#[[1]] & /@ eqEps[[1]] /. q_k_,0 → q_k,i-1) - (#[[1]] & /@ eqEps[[i]]) // Thread

eqOrder[1] /. displayRule
eqOrder[2] /. displayRule
eqOrder[3] /. displayRule
```

$$\{d_0^2 x_0 + x_0\,\omega_1^2 == 0,\ d_0^2 y_0 + y_0\,\omega_2^2 == 0\}$$

$$\{d_0^2 x_1 + x_1\,\omega_1^2 == \mu\, d_0 x_0 - 2\, d_0\, d_1 x_0 + d_0 x_0{}^2\, b_0 - 2\, d_0 x_0\, d_0 y_0\, b_0 + d_0 y_0{}^2\, b_0,$$
$$d_0^2 y_1 + y_1\,\omega_2^2 == \nu\, d_0 x_0 - 2\, d_0\, d_1 y_0 - d_0 x_0{}^2\, b_0 + 2\, d_0 x_0\, d_0 y_0\, b_0 - d_0 y_0{}^2\, b_0 - 2\,\sigma\, y_0\,\omega_2\}$$

$$\{d_0^2 x_2 + x_2\,\omega_1^2 ==$$
$$\mu\, d_0 x_1 + \mu\, d_1 x_0 - 2\, d_0\, d_1 x_1 - d_1^2 x_0 - 2\, d_0\, d_2 x_0 + 2\, d_0 x_0\, d_0 x_1\, b_0 - 2\, d_0 x_1\, d_0 y_0\, b_0 - 2\, d_0 x_0\, d_0 y_1\, b_0 +$$
$$2\, d_0 y_0\, d_0 y_1\, b_0 + 2\, d_0 x_0\, d_1 x_0\, b_0 - 2\, d_0 y_0\, d_1 x_0\, b_0 - 2\, d_0 x_0\, d_1 y_0\, b_0 + 2\, d_0 y_0\, d_1 y_0\, b_0 - d_0 x_0{}^3\, b_1 - c\, x_0^3,$$
$$d_0^2 y_2 + y_2\,\omega_2^2 == \nu\, d_0 x_1 + \nu\, d_1 x_0 - 2\, d_0\, d_1 y_1 - d_1^2 y_0 - 2\, d_0\, d_2 y_0 - 2\, d_0 x_0\, d_0 x_1\, b_0 +$$
$$2\, d_0 x_1\, d_0 y_0\, b_0 + 2\, d_0 x_0\, d_0 y_1\, b_0 - 2\, d_0 y_0\, d_0 y_1\, b_0 - 2\, d_0 x_0\, d_1 x_0\, b_0 + 2\, d_0 y_0\, d_1 x_0\, b_0 +$$
$$2\, d_0 x_0\, d_1 y_0\, b_0 - 2\, d_0 y_0\, d_1 y_0\, b_0 - d_0 y_0{}^3\, b_2 - \sigma^2\, y_0 - c\, y_0^3 - 2\,\sigma\, y_1\,\omega_2\}$$

Resonance condition

```
ResonanceCond = {ω_1 == 1/2 ω_2};
```

Involved frequencies

```
omgList = {ω_1, ω_2};
```

Utility (don't modify)

```
omgRule = Solve[ResonanceCond, {#, #} // Flatten][[1]] & /@ omgList // Reverse
```

$$\left\{\{\omega_2 \to 2\,\omega_1\},\ \left\{\omega_1 \to \frac{\omega_2}{2}\right\}\right\}$$

First-Order Problem

Equations (=0)

```
linearSys = #[[1]] & /@ eqOrder[1];
linearSys /. displayRule // TableForm
```

$$d_0^2 x_0 + x_0\,\omega_1^2$$
$$d_0^2 y_0 + y_0\,\omega_2^2$$

Formal solution of the First-Order Problem

```
sol1 = {q1,0 -> Function[{T0, T1, T2}, A1[T1, T2] Exp[I ω1 T0] + Ā1[T1, T2] Exp[-I ω1 T0]],
   q2,0 -> Function[{T0, T1, T2}, A2[T1, T2] Exp[I ω2 T0] + Ā2[T1, T2] Exp[-I ω2 T0]]}
```

$$\{q_{1,0} \to \mathrm{Function}[\{T_0, T_1, T_2\}, A_1[T_1, T_2]\,\mathrm{Exp}[i\,\omega_1\,T_0] + \bar{A}_1[T_1, T_2]\,\mathrm{Exp}[-i\,\omega_1\,T_0]],$$
$$q_{2,0} \to \mathrm{Function}[\{T_0, T_1, T_2\}, A_2[T_1, T_2]\,\mathrm{Exp}[i\,\omega_2\,T_0] + \bar{A}_2[T_1, T_2]\,\mathrm{Exp}[-i\,\omega_2\,T_0]]\}$$

Second-Order Problem

Substitution of the solution on the Second-Order Problem and representation

```
order2Eq = eqOrder[2] /. sol1 // ExpandAll;
order2Eq /. displayRule
```

$$\Big\{d_0^2 x_1 + x_1\,\omega_1^2 == -2\,i\,e^{i\,T_0\,\omega_1}\,d_1A_1\,\omega_1 + 2\,i\,e^{-i\,T_0\,\omega_1}\,d_1\bar{A}_1\,\omega_1 +$$
$$i\,e^{i\,T_0\,\omega_1}\,\mu\,A_1\,\omega_1 - e^{2\,i\,T_0\,\omega_1}\,A_1^2\,b_0\,\omega_1^2 + 2\,e^{i\,T_0\,\omega_1+i\,T_0\,\omega_2}\,A_1\,A_2\,b_0\,\omega_1\,\omega_2 - e^{2\,i\,T_0\,\omega_2}\,A_2^2\,b_0\,\omega_2^2 -$$
$$i\,e^{-i\,T_0\,\omega_1}\,\mu\,\omega_1\,\bar{A}_1 + 2\,A_1\,b_0\,\omega_1^2\,\bar{A}_1 - 2\,e^{-i\,T_0\,\omega_1+i\,T_0\,\omega_2}\,A_2\,b_0\,\omega_1\,\omega_2\,\bar{A}_1 - e^{-2\,i\,T_0\,\omega_1}\,b_0\,\omega_1^2\,\bar{A}_1^2 -$$
$$2\,e^{i\,T_0\,\omega_1-i\,T_0\,\omega_2}\,A_1\,b_0\,\omega_1\,\omega_2\,\bar{A}_2 + 2\,A_2\,b_0\,\omega_2^2\,\bar{A}_2 + 2\,e^{-i\,T_0\,\omega_1-i\,T_0\,\omega_2}\,b_0\,\omega_1\,\omega_2\,\bar{A}_1\,\bar{A}_2 - e^{-2\,i\,T_0\,\omega_2}\,b_0\,\omega_2^2\,\bar{A}_2^2,$$
$$d_0^2 y_1 + y_1\,\omega_2^2 == i\,e^{i\,T_0\,\omega_1}\,\nu\,A_1\,\omega_1 + e^{2\,i\,T_0\,\omega_1}\,A_1^2\,b_0\,\omega_1^2 - 2\,i\,e^{i\,T_0\,\omega_2}\,d_1A_2\,\omega_2 + 2\,i\,e^{-i\,T_0\,\omega_2}\,d_1\bar{A}_2\,\omega_2 -$$
$$2\,e^{i\,T_0\,\omega_2}\,\sigma\,A_2\,\omega_2 - 2\,e^{i\,T_0\,\omega_1+i\,T_0\,\omega_2}\,A_1\,A_2\,b_0\,\omega_1\,\omega_2 + e^{2\,i\,T_0\,\omega_2}\,A_2^2\,b_0\,\omega_2^2 - i\,e^{-i\,T_0\,\omega_1}\,\nu\,\omega_1\,\bar{A}_1 -$$
$$2\,A_1\,b_0\,\omega_1^2\,\bar{A}_1 + 2\,e^{-i\,T_0\,\omega_1+i\,T_0\,\omega_2}\,A_2\,b_0\,\omega_1\,\omega_2\,\bar{A}_1 + e^{-2\,i\,T_0\,\omega_1}\,b_0\,\omega_1^2\,\bar{A}_1^2 - 2\,e^{-i\,T_0\,\omega_2}\,\sigma\,\omega_2\,\bar{A}_2 +$$
$$2\,e^{i\,T_0\,\omega_1-i\,T_0\,\omega_2}\,A_1\,b_0\,\omega_1\,\omega_2\,\bar{A}_2 - 2\,A_2\,b_0\,\omega_2^2\,\bar{A}_2 - 2\,e^{-i\,T_0\,\omega_1-i\,T_0\,\omega_2}\,b_0\,\omega_1\,\omega_2\,\bar{A}_1\,\bar{A}_2 + e^{-2\,i\,T_0\,\omega_2}\,b_0\,\omega_2^2\,\bar{A}_2^2\Big\}$$

Utility (don't modify)

```
expRule1[i_] := Exp[a_] :> Exp[Expand[a /. omgRule[[i]]] /. ϵ T0 -> T1]
```

Terms of type $e^{i\,\omega_1\,T_0}$ in the first equation of the Second-Order Problem and representation

```
ST11 = Coefficient[order2Eq[[#, 2]] /. expRule1[1], Exp[I ω1 T0]] & /@ {1};
ST11 /. displayRule
```

$$\left\{-2\,i\,d_1A_1\,\omega_1+i\,\mu\,A_1\,\omega_1-2\,A_2\,b_0\,\omega_1\,\omega_2\,\overline{A}_1\right\}$$

Terms of type $e^{i\,\omega_2\,T_0}$ in the first equation of the Second-Order Problem and representation

```
ST12 = Coefficient[order2Eq[[#, 2]] /. expRule1[2], Exp[I ω2 T0]] & /@ {1};
ST12 /. displayRule
```

$$\left\{-A_1^2\,b_0\,\omega_1^2\right\}$$

Terms of type $e^{i\,\omega_1\,T_0}$ in the second equation of the Second-Order Problem and representation

```
ST21 = Coefficient[order2Eq[[#, 2]] /. expRule1[1], Exp[I ω1 T0]] & /@ {2};
ST21 /. displayRule
```

$$\left\{i\,\nu\,A_1\,\omega_1+2\,A_2\,b_0\,\omega_1\,\omega_2\,\overline{A}_1\right\}$$

Terms of type $e^{i\,\omega_2\,T_0}$ in the second equation of the Second-Order Problem and representation

```
ST22 = Coefficient[order2Eq[[#, 2]] /. expRule1[2], Exp[I ω2 T0]] & /@ {2};
ST22 /. displayRule
```

$$\left\{A_1^2\,b_0\,\omega_1^2-2\,i\,d_1A_2\,\omega_2-2\,\sigma\,A_2\,\omega_2\right\}$$

Scalar product with the left eigenvectors: First-Order AME; representation

```
SCond1 = {{1, 0}.{ST11, ST21} == 0, {0, 1}.{ST12, ST22} == 0};
SCond1 /. displayRule
```

$$\left\{\left\{-2\,i\,d_1A_1\,\omega_1+i\,\mu\,A_1\,\omega_1-2\,A_2\,b_0\,\omega_1\,\omega_2\,\overline{A}_1\right\}=0,\ \left\{A_1^2\,b_0\,\omega_1^2-2\,i\,d_1A_2\,\omega_2-2\,\sigma\,A_2\,\omega_2\right\}=0\right\}$$

Algebraic manipulation to obtain $D_1\,A_1$ and $D_1\,A_2$

```
SCond1Rule1 = Solve[SCond1, {A1^(1,0)[T1, T2], A2^(1,0)[T1, T2]}][[1]] // ExpandAll;
SCond1Rule1 /. displayRule // TableForm
```

$$d_1A_1\rightarrow\frac{\mu\,A_1}{2}+i\,A_2\,b_0\,\omega_2\,\overline{A}_1$$

$$d_1A_2\rightarrow i\,\sigma\,A_2-\frac{i\,A_1^2\,b_0\,\omega_1^2}{2\,\omega_2}$$

Substitution of the First - Order AME to the Second Order Equations

```
order2Eqm =
  (((order2Eq[[#]] /. SCond1Rule1 /. (SCond1Rule1 /. conjugateRule) /. expRule1[#]) & /@
      {1, 2}) // ExpandAll); order2Eqm /. displayRule
```

$$\left\{d_0^2 x_1 + x_1\,\omega_1^2 = -e^{2\,i\,T_0\,\omega_1} A_1^2\, b_0\,\omega_1^2 + 2\,e^{3\,i\,T_0\,\omega_1} A_1\, A_2\, b_0\,\omega_1\,\omega_2 - e^{4\,i\,T_0\,\omega_1} A_2^2\, b_0\,\omega_2^2 + 2\,A_1\, b_0\,\omega_1^2\,\bar{A}_1 - e^{-2\,i\,T_0\,\omega_1} b_0\,\omega_1^2\,\bar{A}_1^2 + 2\,A_2\, b_0\,\omega_2^2\,\bar{A}_2 + 2\,e^{-3\,i\,T_0\,\omega_1} b_0\,\omega_1\,\omega_2\,\bar{A}_1\,\bar{A}_2 - e^{-4\,i\,T_0\,\omega_1} b_0\,\omega_2^2\,\bar{A}_2^2,\right.$$
$$d_0^2 y_1 + y_1\,\omega_2^2 = i\,e^{\frac{1}{2}\,i\,T_0\,\omega_2}\,\nu\, A_1\,\omega_1 - 2\,e^{\frac{3}{2}\,i\,T_0\,\omega_2} A_1\, A_2\, b_0\,\omega_1\,\omega_2 + e^{2\,i\,T_0\,\omega_2} A_2^2\, b_0\,\omega_2^2 - i\,e^{-\frac{1}{2}\,i\,T_0\,\omega_2}\,\nu\,\omega_1\,\bar{A}_1 - 2\,A_1\, b_0\,\omega_1^2\,\bar{A}_1 + 2\,e^{\frac{1}{2}\,i\,T_0\,\omega_2} A_2\, b_0\,\omega_1\,\omega_2\,\bar{A}_1 +$$
$$\left. 2\,e^{-\frac{1}{2}\,i\,T_0\,\omega_2} A_1\, b_0\,\omega_1\,\omega_2\,\bar{A}_2 - 2\,A_2\, b_0\,\omega_2^2\,\bar{A}_2 - 2\,e^{-\frac{3}{2}\,i\,T_0\,\omega_2} b_0\,\omega_1\,\omega_2\,\bar{A}_1\,\bar{A}_2 + e^{-2\,i\,T_0\,\omega_2} b_0\,\omega_2^2\,\bar{A}_2^2\right\}$$

Assumpion on the coefficients

```
$Assumptions = {ω1, ω2, b0, c, μ, ν, b1, b2, σ} ∈ Reals
```

$(\omega_1 \mid \omega_2 \mid b_0 \mid c \mid \mu \mid \nu \mid b_1 \mid b_2 \mid \sigma) \in \text{Reals}$

Construction of the particular solution of the first equation of the Second-Order Problem and representation

```
solOrder2Eqm[1] =
  Simplify[Table[q1,1[T0, T1, T2] /. TrigToExp[Flatten[DSolve[{order2Eqm[[1, 1]] ==
          order2Eqm[[1, 2, i]]}, q1,1[T0, T1, T2], T0] /. {C[1] -> 0, C[2] -> 0}]],
    {i, 1, Length[order2Eqm[[1, 2]]]}]]; solOrder2Eqm[1] /. displayRule
```

$$\left\{\frac{1}{3}\,e^{2\,i\,T_0\,\omega_1} A_1^2\, b_0,\ -\frac{e^{3\,i\,T_0\,\omega_1} A_1\, A_2\, b_0\,\omega_2}{4\,\omega_1},\ \frac{e^{4\,i\,T_0\,\omega_1} A_2^2\, b_0\,\omega_2^2}{15\,\omega_1^2},\ 2\,A_1\, b_0\,\bar{A}_1,\ \frac{1}{3}\,e^{-2\,i\,T_0\,\omega_1} b_0\,\bar{A}_1^2,\ \frac{2\,A_2\, b_0\,\omega_2^2\,\bar{A}_2}{\omega_1^2},\ -\frac{e^{-3\,i\,T_0\,\omega_1} b_0\,\omega_2\,\bar{A}_1\,\bar{A}_2}{4\,\omega_1},\ \frac{e^{-4\,i\,T_0\,\omega_1} b_0\,\omega_2^2\,\bar{A}_2^2}{15\,\omega_1^2}\right\}$$

```
solq11 = Expand[Total[solOrder2Eqm[1]]];
solq11 /. displayRule
```

$$\frac{1}{3}\,e^{2\,i\,T_0\,\omega_1} A_1^2\, b_0 - \frac{e^{3\,i\,T_0\,\omega_1} A_1\, A_2\, b_0\,\omega_2}{4\,\omega_1} + \frac{e^{4\,i\,T_0\,\omega_1} A_2^2\, b_0\,\omega_2^2}{15\,\omega_1^2} + 2\,A_1\, b_0\,\bar{A}_1 + \frac{1}{3}\,e^{-2\,i\,T_0\,\omega_1} b_0\,\bar{A}_1^2 + \frac{2\,A_2\, b_0\,\omega_2^2\,\bar{A}_2}{\omega_1^2} - \frac{e^{-3\,i\,T_0\,\omega_1} b_0\,\omega_2\,\bar{A}_1\,\bar{A}_2}{4\,\omega_1} + \frac{e^{-4\,i\,T_0\,\omega_1} b_0\,\omega_2^2\,\bar{A}_2^2}{15\,\omega_1^2}$$

Construction of the particular solution of the second equation of the Second-Order Problem

```
solOrder2Eqm[2] = Table[q2,1[T0, T1, T2] /.
    Simplify[TrigToExp[Flatten[DSolve[{order2Eqm[[2, 1]] == order2Eqm[[2, 2, i]]},
          q2,1[T0, T1, T2], T0] /. {C[1] -> 0, C[2] -> 0}]]],
   {i, 1, Length[order2Eqm[[2, 2]]]}]; solOrder2Eqm[2] /. displayRule
```

$$\left\{\frac{4\,i\,e^{\frac{1}{2} i T_0 \omega_2}\,\nu\,A_1\,\omega_1}{3\,\omega_2^2},\ \frac{8\,e^{\frac{3}{2} i T_0 \omega_2}\,A_1\,A_2\,b_0\,\omega_1}{5\,\omega_2},\ -\frac{1}{3}\,e^{2 i T_0 \omega_2}\,A_2^2\,b_0,\right.$$
$$-\frac{4\,i\,e^{-\frac{1}{2} i T_0 \omega_2}\,\nu\,\omega_1\,\bar{A}_1}{3\,\omega_2^2},\ -\frac{2\,A_1\,b_0\,\omega_1^2\,\bar{A}_1}{\omega_2^2},\ \frac{8\,e^{\frac{1}{2} i T_0 \omega_2}\,A_2\,b_0\,\omega_1\,\bar{A}_1}{3\,\omega_2},$$
$$\left.\frac{8\,e^{-\frac{1}{2} i T_0 \omega_2}\,A_1\,b_0\,\omega_1\,\bar{A}_2}{3\,\omega_2},\ -2\,A_2\,b_0\,\bar{A}_2,\ \frac{8\,e^{-\frac{3}{2} i T_0 \omega_2}\,b_0\,\omega_1\,\bar{A}_1\,\bar{A}_2}{5\,\omega_2},\ -\frac{1}{3}\,e^{-2 i T_0 \omega_2}\,b_0\,\bar{A}_2^2\right\}$$

```
solq12 = Expand[Total[solOrder2Eqm[2]]];
solq12 /. displayRule
```

$$-\frac{1}{3}\,e^{2 i T_0 \omega_2}\,A_2^2\,b_0+\frac{4\,i\,e^{\frac{1}{2} i T_0 \omega_2}\,\nu\,A_1\,\omega_1}{3\,\omega_2^2}+\frac{8\,e^{\frac{3}{2} i T_0 \omega_2}\,A_1\,A_2\,b_0\,\omega_1}{5\,\omega_2}-\frac{4\,i\,e^{-\frac{1}{2} i T_0 \omega_2}\,\nu\,\omega_1\,\bar{A}_1}{3\,\omega_2^2}-\frac{2\,A_1\,b_0\,\omega_1^2\,\bar{A}_1}{\omega_2^2}+$$
$$\frac{8\,e^{\frac{1}{2} i T_0 \omega_2}\,A_2\,b_0\,\omega_1\,\bar{A}_1}{3\,\omega_2}-2\,A_2\,b_0\,\bar{A}_2+\frac{8\,e^{-\frac{1}{2} i T_0 \omega_2}\,A_1\,b_0\,\omega_1\,\bar{A}_2}{3\,\omega_2}+\frac{8\,e^{-\frac{3}{2} i T_0 \omega_2}\,b_0\,\omega_1\,\bar{A}_1\,\bar{A}_2}{5\,\omega_2}-\frac{1}{3}\,e^{-2 i T_0 \omega_2}\,b_0\,\bar{A}_2^2$$

Formal representation of the solution

```
sol2 = {q1,1 -> Function[{T0, T1, T2}, Evaluate[solq11] ],
    q2,1 -> Function[{T0, T1, T2}, Evaluate[solq12]]};
sol2 /. displayRule
```

$$\left\{q_{1,1}\to \text{Function}\left[\{T_0, T_1, T_2\},\ \frac{1}{3}\,e^{2 i T_0 \omega_1}\,b_0\,A_1^2-\frac{e^{3 i T_0 \omega_1}\,b_0\,\omega_2\,A_1\,A_2}{4\,\omega_1}+\frac{e^{4 i T_0 \omega_1}\,b_0\,\omega_2^2\,A_2^2}{15\,\omega_1^2}+\right.\right.$$
$$\left.2\,b_0\,A_1\,\bar{A}_1+\frac{1}{3}\,e^{-2 i T_0 \omega_1}\,b_0\,\bar{A}_1^2+\frac{2\,b_0\,\omega_2^2\,A_2\,\bar{A}_2}{\omega_1^2}-\frac{e^{-3 i T_0 \omega_1}\,b_0\,\omega_2\,\bar{A}_1\,\bar{A}_2}{4\,\omega_1}+\frac{e^{-4 i T_0 \omega_1}\,b_0\,\omega_2^2\,\bar{A}_2^2}{15\,\omega_1^2}\right],$$
$$q_{2,1}\to \text{Function}\left[\{T_0, T_1, T_2\},\ \frac{4\,i\,e^{\frac{1}{2} i T_0 \omega_2}\,\nu\,\omega_1\,A_1}{3\,\omega_2^2}+\frac{8\,e^{\frac{3}{2} i T_0 \omega_2}\,b_0\,\omega_1\,A_1\,A_2}{5\,\omega_2}-\right.$$
$$\frac{1}{3}\,e^{2 i T_0 \omega_2}\,b_0\,A_2^2-\frac{4\,i\,e^{-\frac{1}{2} i T_0 \omega_2}\,\nu\,\omega_1\,\bar{A}_1}{3\,\omega_2^2}-\frac{2\,b_0\,\omega_1^2\,A_1\,\bar{A}_1}{\omega_2^2}+\frac{8\,e^{\frac{1}{2} i T_0 \omega_2}\,b_0\,\omega_1\,A_2\,\bar{A}_1}{3\,\omega_2}+$$
$$\left.\left.\frac{8\,e^{-\frac{1}{2} i T_0 \omega_2}\,b_0\,\omega_1\,A_1\,\bar{A}_2}{3\,\omega_2}-2\,b_0\,A_2\,\bar{A}_2+\frac{8\,e^{-\frac{3}{2} i T_0 \omega_2}\,b_0\,\omega_1\,\bar{A}_1\,\bar{A}_2}{5\,\omega_2}-\frac{1}{3}\,e^{-2 i T_0 \omega_2}\,b_0\,\bar{A}_2^2\right]\right\}$$

Third-Order Problem

Substitution in the Third-Order Equations

```
order3Eq = eqOrder[3] /. sol1 /. sol2 // ExpandAll;
```

Simplifications

```
order3Eqpr[1, 2] = Simplify[order3Eq[[1, 2]]];
```

```
order3Eqpr[2, 2] = Simplify[order3Eq[[2, 2]]];
```

Terms of type $e^{i\,\omega_1\,T_0}$ in the first equation of the Third-Order Problem and representation

```
ST311 = Coefficient[order3Eqpr[#, 2] /. expRule1[#], Exp[I ω1 T0]] & /@ {1};
ST311 /. displayRule
```

$$\left\{\frac{1}{60\,\omega_1\,\omega_2}\left(60\,\mu\,d_1A_1\,\omega_1\,\omega_2 - 60\,d_1^2A_1\,\omega_1\,\omega_2 - 120\,i\,d_2A_1\,\omega_1^2\,\omega_2 - 120\,i\,d_1\overline{A}_1\,A_2\,b_0\,\omega_1\,\omega_2^2 - 180\,c\,A_1^2\,\omega_1\,\omega_2\,\overline{A}_1 + 120\,i\,d_1A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 - 80\,i\,\nu\,A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 + 80\,A_1^2\,b_0^2\,\omega_1^3\,\omega_2\,\overline{A}_1 - 180\,i\,A_1^2\,b_1\,\omega_1^4\,\omega_2\,\overline{A}_1 + 448\,A_1\,A_2\,b_0^2\,\omega_1^2\,\omega_2^2\,\overline{A}_2 + 90\,A_1\,A_2\,b_0^2\,\omega_1\,\omega_2^3\,\overline{A}_2\right)\right\}$$

Terms of type $e^{i\,\omega_2\,T_0}$ in the first equation of the Third-Order Problem and representation

```
ST312 = Coefficient[order3Eqpr[#, 2] /. expRule1[#], Exp[I ω2 T0]] & /@ {1};
ST312 /. displayRule
```

{0}

Terms of type $e^{i\,\omega_1\,T_0}$ in the second equation of the Third-Order Problem and representation

```
ST321 = Coefficient[order3Eqpr[#, 2] /. expRule1[1], Exp[I ω1 T0]] & /@ {2};
ST321 /. displayRule
```

$$\left\{\frac{1}{60\,\omega_1\,\omega_2}\left(80\,\nu\,d_1A_1\,\omega_1^2 - 160\,i\,\nu\,\sigma\,A_1\,\omega_1^2 + 60\,\nu\,d_1A_1\,\omega_1\,\omega_2 - 160\,i\,d_1\overline{A}_1\,A_2\,b_0\,\omega_1^2\,\omega_2 + 120\,i\,d_1\overline{A}_1\,A_2\,b_0\,\omega_1\,\omega_2^2 - 280\,i\,d_1A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 + 80\,i\,\nu\,A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 - 320\,\sigma\,A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 - 80\,A_1^2\,b_0^2\,\omega_1^3\,\omega_2\,\overline{A}_1 - 448\,A_1\,A_2\,b_0^2\,\omega_1^2\,\omega_2^2\,\overline{A}_2 - 90\,A_1\,A_2\,b_0^2\,\omega_1\,\omega_2^3\,\overline{A}_2\right)\right\}$$

Terms of type $e^{i\,\omega_2\,T_0}$ in the second equation of the Third-Order Problem and representation

```
ST322 = Coefficient[order3Eqpr[#, 2] /. expRule1[2], Exp[I ω2 T0]] & /@ {2};
ST322 /. displayRule
```

$$\left\{\frac{1}{60\,\omega_1\,\omega_2}\left(-80\,i\,\nu\,A_1^2\,b_0\,\omega_1^3 - 60\,d_1^2A_2\,\omega_1\,\omega_2 - 60\,\sigma^2\,A_2\,\omega_1\,\omega_2 - 120\,i\,d_1A_1\,A_1\,b_0\,\omega_1^2\,\omega_2 + 40\,i\,\nu\,A_1^2\,b_0\,\omega_1^2\,\omega_2 - 120\,i\,d_2A_2\,\omega_1\,\omega_2^2 + 128\,A_1\,A_2\,b_0^2\,\omega_1^3\,\omega_2\,\overline{A}_1 + 90\,A_1\,A_2\,b_0^2\,\omega_1^2\,\omega_2^2\,\overline{A}_1 - 180\,c\,A_2^2\,\omega_1\,\omega_2\,\overline{A}_2 + 80\,A_2^2\,b_0^2\,\omega_1\,\omega_2^3\,\overline{A}_2 + 32\,A_2^2\,b_0^2\,\omega_2^4\,\overline{A}_2 - 180\,i\,A_2^2\,b_2\,\omega_1\,\omega_2^4\,\overline{A}_2\right)\right\}$$

Scalar product with the left eigenvectors: Second-Order AME; representation

```
SCond2 = {{1, 0}.{ST311, ST321} == 0, {0, 1}.{ST312, ST322} == 0};
SCond2 /. displayRule
```

$$\left\{\left\{\frac{1}{60\,\omega_1\,\omega_2}\left(60\,\mu\,d_1A_1\,\omega_1\,\omega_2 - 60\,d_1^2A_1\,\omega_1\,\omega_2 - 120\,i\,d_2A_1\,\omega_1^2\,\omega_2 - 120\,i\,d_1\overline{A}_1\,A_2\,b_0\,\omega_1\,\omega_2^2 - 180\,c\,A_1^2\,\omega_1\,\omega_2\,\overline{A}_1 + 120\,i\,d_1A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 - 80\,i\,\nu\,A_2\,b_0\,\omega_1^2\,\omega_2\,\overline{A}_1 + 80\,A_1^2\,b_0^2\,\omega_1^3\,\omega_2\,\overline{A}_1 - 180\,i\,A_1^2\,b_1\,\omega_1^4\,\omega_2\,\overline{A}_1 + 448\,A_1\,A_2\,b_0^2\,\omega_1^2\,\omega_2^2\,\overline{A}_2 + 90\,A_1\,A_2\,b_0^2\,\omega_1\,\omega_2^3\,\overline{A}_2\right)\right\} == 0,\right.$$

$$\left.\left\{\frac{1}{60\,\omega_1\,\omega_2}\left(-80\,i\,\nu\,A_1^2\,b_0\,\omega_1^3 - 60\,d_1^2A_2\,\omega_1\,\omega_2 - 60\,\sigma^2\,A_2\,\omega_1\,\omega_2 - 120\,i\,d_1A_1\,A_1\,b_0\,\omega_1^2\,\omega_2 + 40\,i\,\nu\,A_1^2\,b_0\,\omega_1^2\,\omega_2 - 120\,i\,d_2A_2\,\omega_1\,\omega_2^2 + 128\,A_1\,A_2\,b_0^2\,\omega_1^3\,\omega_2\,\overline{A}_1 + 90\,A_1\,A_2\,b_0^2\,\omega_1^2\,\omega_2^2\,\overline{A}_1 - 180\,c\,A_2^2\,\omega_1\,\omega_2\,\overline{A}_2 + 80\,A_2^2\,b_0^2\,\omega_1\,\omega_2^3\,\overline{A}_2 + 32\,A_2^2\,b_0^2\,\omega_2^4\,\overline{A}_2 - 180\,i\,A_2^2\,b_2\,\omega_1\,\omega_2^4\,\overline{A}_2\right)\right\} == 0\right\}$$

Algebraic manipulation to obtain $D_2\,A_1$ and $D_2\,A_2$

```
SCond2Rule1 =
  Solve[SCond2, {A1^(0,1)[T1, T2], A2^(0,1)[T1, T2]}][[1]] /. (A1^(2,0)[T1, T2] → ∂T1 SCond1Rule1[[
        1, 2]]) /. (A2^(2,0)[T1, T2] → ∂T1 SCond1Rule1[[2, 2]]) /. SCond1Rule1 /.
     (SCond1Rule1 /. conjugateRule) // ExpandAll // Simplify // Expand;
SCond2Rule1 /. displayRule
```

$$\left\{d_2A_1 \to -\frac{i\,\mu^2\,A_1}{8\,\omega_1} - \frac{2}{3}\,\nu\,A_2\,b_0\,\overline{A}_1 + i\,\sigma\,A_2\,b_0\,\overline{A}_1 + \frac{3\,i\,c\,A_1^2\,\overline{A}_1}{2\,\omega_1} - \frac{5}{12}\,i\,A_1^2\,b_0^2\,\omega_1\,\overline{A}_1 - \frac{3}{2}\,A_1^2\,b_1\,\omega_1^2\,\overline{A}_1 - \frac{i\,A_1^2\,b_0^2\,\omega_1^2\,\overline{A}_1}{2\,\omega_2} - \frac{\mu\,A_2\,b_0\,\omega_2\,\overline{A}_1}{2\,\omega_1} - \frac{i\,\sigma\,A_2\,b_0\,\omega_2\,\overline{A}_1}{2\,\omega_1} - \frac{56}{15}\,i\,A_1\,A_2\,b_0^2\,\omega_2\,\overline{A}_2 + \frac{3\,i\,A_1\,A_2\,b_0^2\,\omega_2^2\,\overline{A}_2}{4\,\omega_1},\right.$$

$$\left.d_2A_2 \to \frac{\mu\,A_1^2\,b_0\,\omega_1^2}{4\,\omega_2^2} - \frac{2\,\nu\,A_1^2\,b_0\,\omega_1^2}{3\,\omega_2^2} + \frac{i\,\sigma\,A_1^2\,b_0\,\omega_1^2}{4\,\omega_2^2} - \frac{\mu\,A_1^2\,b_0\,\omega_1}{2\,\omega_2} + \frac{\nu\,A_1^2\,b_0\,\omega_1}{3\,\omega_2} - \frac{7}{4}\,i\,A_1\,A_2\,b_0^2\,\omega_1\,\overline{A}_1 - \frac{17\,i\,A_1\,A_2\,b_0^2\,\omega_1^2\,\overline{A}_1}{30\,\omega_2} + \frac{3\,i\,c\,A_2^2\,\overline{A}_2}{2\,\omega_2} - \frac{2}{3}\,i\,A_2^2\,b_0^2\,\omega_2\,\overline{A}_2 - \frac{3}{2}\,A_2^2\,b_2\,\omega_2^2\,\overline{A}_2 - \frac{4\,i\,A_2^2\,b_0^2\,\omega_2^2\,\overline{A}_2}{15\,\omega_1}\right\}$$

Reconstitution of the AME and of the solution

$\frac{dA_1}{dt}$ = ame[1]

```
ame[1] = (A1^(1,0)[T1, T2] + A1^(0,1)[T1, T2]) /.
   Join[SCond1Rule1, SCond2Rule1 /. omgRule[[1]]]; ame[1] /. displayRule
```

$$\frac{\mu\,A_1}{2} - \frac{i\,\mu^2\,A_1}{8\,\omega_1} - \mu\,A_2\,b_0\,\overline{A}_1 - \frac{2}{3}\,\nu\,A_2\,b_0\,\overline{A}_1 + \frac{3\,i\,c\,A_1^2\,\overline{A}_1}{2\,\omega_1} - \frac{2}{3}\,i\,A_1^2\,b_0^2\,\omega_1\,\overline{A}_1 - \frac{3}{2}\,A_1^2\,b_1\,\omega_1^2\,\overline{A}_1 + i\,A_2\,b_0\,\omega_2\,\overline{A}_1 - \frac{67}{15}\,i\,A_1\,A_2\,b_0^2\,\omega_1\,\overline{A}_2$$

$\frac{dA_2}{dt}$ = ame[2]

```
ame[2] = (A2^(1,0)[T1, T2] + A2^(0,1)[T1, T2]) /.
  Join[SCond1Rule1, SCond2Rule1 /. omgRule[[1]]]; ame[2] /. displayRule
```

$$i\,\sigma\,A_2 - \frac{3}{16}\,\mu\,A_1^2\,b_0 + \frac{1}{16}\,i\,\sigma\,A_1^2\,b_0 - \frac{i\,A_1^2\,b_0\,\omega_1^2}{2\,\omega_2} - \frac{61}{30}\,i\,A_1\,A_2\,b_0^2\,\omega_1\,\bar{A}_1 + \frac{3\,i\,c\,A_2^2\,\bar{A}_2}{4\,\omega_1} - \frac{12}{5}\,i\,A_2^2\,b_0^2\,\omega_1\,\bar{A}_2 - 6\,A_2^2\,b_2\,\omega_1^2\,\bar{A}_2$$

Better representation

```
Collect[ame[1], {A1[T1, T2], A1[T1, T2]^2 Ā1[T1, T2],
   A1[T1, T2] A2[T1, T2] Ā2[T1, T2], A2[T1, T2] Ā1[T1, T2]}] /. displayRule
```

$$A_1\left(\frac{\mu}{2} - \frac{i\,\mu^2}{8\,\omega_1}\right) + A_1^2\left(\frac{3\,i\,c}{2\,\omega_1} - \frac{2}{3}\,i\,b_0^2\,\omega_1 - \frac{3}{2}\,b_1\,\omega_1^2\right)\bar{A}_1 + A_2\left(-\mu\,b_0 - \frac{2\,\nu\,b_0}{3} + i\,b_0\,\omega_2\right)\bar{A}_1 - \frac{67}{15}\,i\,A_1\,A_2\,b_0^2\,\omega_1\,\bar{A}_2$$

```
Collect[ame[2],
  {A2[T1, T2], A1[T1, T2]^2, A1[T1, T2]^2 Ā1[T1, T2], A1[T1, T2] A2[T1, T2] Ā2[T1, T2],
   A1[T1, T2] A2[T1, T2] Ā1[T1, T2], A2[T1, T2]^2 Ā2[T1, T2]}] /. displayRule
```

$$i\,\sigma\,A_2 + A_1^2\left(-\frac{3\,\mu\,b_0}{16} + \frac{1}{16}\,i\,\sigma\,b_0 - \frac{i\,b_0\,\omega_1^2}{2\,\omega_2}\right) - \frac{61}{30}\,i\,A_1\,A_2\,b_0^2\,\omega_1\,\bar{A}_1 + A_2^2\left(\frac{3\,i\,c}{4\,\omega_1} - \frac{12}{5}\,i\,b_0^2\,\omega_1 - 6\,b_2\,\omega_1^2\right)\bar{A}_2$$

Polar form of the AME

Utility

```
traspRule =
 {A1[T1, T2] -> A1[t], Ā1[T1, T2] -> Ā1[t], A2[T1, T2] -> A2[t], Ā2[T1, T2] -> Ā2[t]}
```

$\{A_1[T_1, T_2] \to A_1[t],\ \bar{A}_1[T_1, T_2] \to \bar{A}_1[t],\ A_2[T_1, T_2] \to A_2[t],\ \bar{A}_2[T_1, T_2] \to \bar{A}_2[t]\}$

Rewriting of the AME

```
amem[1] = (ame[1] /. traspRule) - A1'[t]
```

$$\frac{1}{2}\,\mu\,A_1[t] - \frac{i\,\mu^2\,A_1[t]}{8\,\omega_1} + \frac{3\,i\,c\,A_1[t]^2\,\bar{A}_1[t]}{2\,\omega_1} - \frac{2}{3}\,i\,b_0^2\,\omega_1\,A_1[t]^2\,\bar{A}_1[t] - \frac{3}{2}\,b_1\,\omega_1^2\,A_1[t]^2\,\bar{A}_1[t] - \mu\,b_0\,A_2[t]\,\bar{A}_1[t] - \frac{2}{3}\,\nu\,b_0\,A_2[t]\,\bar{A}_1[t] + i\,b_0\,\omega_2\,A_2[t]\,\bar{A}_1[t] - \frac{67}{15}\,i\,b_0^2\,\omega_1\,A_1[t]\,A_2[t]\,\bar{A}_2[t] - (A_1)'[t]$$

```
amem[2] = (ame[2] /. traspRule) - A2'[t]
```

$$-\frac{3}{16}\mu\, b_0\, A_1[t]^2+\frac{1}{16}\,i\,\sigma\, b_0\, A_1[t]^2-\frac{i\, b_0\,\omega_1^2 A_1[t]^2}{2\,\omega_2}+i\,\sigma A_2[t]-\frac{61}{30}\,i\, b_0^2\,\omega_1 A_1[t]\, A_2[t]\,\overline{A}_1[t]+\frac{3\,i\, c\, A_2[t]^2\,\overline{A}_2[t]}{4\,\omega_1}-\frac{12}{5}\,i\, b_0^2\,\omega_1 A_2[t]^2\,\overline{A}_2[t]-6\, b_2\,\omega_1^2 A_2[t]^2\,\overline{A}_2[t]-(A_2)'[t]$$

Polar form of the complex amplitudes

```
polRule = {A1[t] -> 1/2 a1[t] e^(i ϑ1[t]),
  A2[t] -> 1/2 a2[t] e^(i ϑ2[t]), Ā1[t] -> 1/2 a1[t] e^(-i ϑ1[t]), Ā2[t] -> 1/2 a2[t] e^(-i ϑ2[t])}
```

$$\left\{A_1[t]\to\frac{1}{2}e^{i\vartheta 1[t]}\,a1[t],\ A_2[t]\to\frac{1}{2}e^{i\vartheta 2[t]}\,a2[t],\ \overline{A}_1[t]\to\frac{1}{2}e^{-i\vartheta 1[t]}\,a1[t],\ \overline{A}_2[t]\to\frac{1}{2}e^{-i\vartheta 2[t]}\,a2[t]\right\}$$

```
polRuleD = Join[polRule, D[polRule, t]]
```

$$\left\{A_1[t]\to\frac{1}{2}e^{i\vartheta 1[t]}\,a1[t],\ A_2[t]\to\frac{1}{2}e^{i\vartheta 2[t]}\,a2[t],\ \overline{A}_1[t]\to\frac{1}{2}e^{-i\vartheta 1[t]}\,a1[t],\ \overline{A}_2[t]\to\frac{1}{2}e^{-i\vartheta 2[t]}\,a2[t],\ (A_1)'[t]\to\frac{1}{2}e^{i\vartheta 1[t]}\,a1'[t]+\frac{1}{2}\,i\,e^{i\vartheta 1[t]}\,a1[t]\,\vartheta 1[t],\ (A_2)'[t]\to\frac{1}{2}e^{i\vartheta 2[t]}\,a2'[t]+\frac{1}{2}\,i\,e^{i\vartheta 2[t]}\,a2[t]\,\vartheta 2[t],\ (\overline{A}_1)'[t]\to\frac{1}{2}e^{-i\vartheta 1[t]}\,a1'[t]-\frac{1}{2}\,i\,e^{-i\vartheta 1[t]}\,a1[t]\,\vartheta 1[t],\ (\overline{A}_2)'[t]\to\frac{1}{2}e^{-i\vartheta 2[t]}\,a2'[t]-\frac{1}{2}\,i\,e^{-i\vartheta 2[t]}\,a2[t]\,\vartheta 2[t]\right\}$$

Substitution of the complex amplitudes with the polar form

```
eq[1] = amem[1] /. polRuleD
```

$$\frac{1}{4}e^{i\vartheta 1[t]}\,\mu\, a1[t]-\frac{1}{4}e^{-i\vartheta 1[t]+i\vartheta 2[t]}\,\mu\, a1[t]\, a2[t]\, b_0-\frac{1}{6}e^{-i\vartheta 1[t]+i\vartheta 2[t]}\,\nu\, a1[t]\, a2[t]\, b_0-\frac{i\,e^{i\vartheta 1[t]}\,\mu^2\, a1[t]}{16\,\omega_1}+\frac{3\,i\,c\,e^{i\vartheta 1[t]}\,a1[t]^3}{16\,\omega_1}-\frac{1}{12}\,i\,e^{i\vartheta 1[t]}\,a1[t]^3\, b_0^2\,\omega_1-\frac{67}{120}\,i\,e^{i\vartheta 1[t]}\,a1[t]\, a2[t]^2\, b_0^2\,\omega_1-\frac{3}{16}e^{i\vartheta 1[t]}\,a1[t]^3\, b_1\,\omega_1^2+\frac{1}{4}\,i\,e^{-i\vartheta 1[t]+i\vartheta 2[t]}\,a1[t]\, a2[t]\, b_0\,\omega_2-\frac{1}{2}e^{i\vartheta 1[t]}\,a1'[t]-\frac{1}{2}\,i\,e^{i\vartheta 1[t]}\,a1[t]\,\vartheta 1[t]$$

```
eq[2] = amem[2] /. polRuleD
```

$$\frac{1}{2} i e^{i \vartheta 2[t]} \sigma a2[t] - \frac{3}{64} e^{2 i \vartheta 1[t]} \mu a1[t]^2 b_0 + \frac{1}{64} i e^{2 i \vartheta 1[t]} \sigma a1[t]^2 b_0 + \frac{3 i c e^{i \vartheta 2[t]} a2[t]^3}{32 \omega_1} - \frac{61}{240} i e^{i \vartheta 2[t]} a1[t]^2 a2[t] b_0^2 \omega_1 - \frac{3}{10} i e^{i \vartheta 2[t]} a2[t]^3 b_0^2 \omega_1 - \frac{3}{4} e^{i \vartheta 2[t]} a2[t]^3 b_2 \omega_1^2 - \frac{i e^{2 i \vartheta 1[t]} a1[t]^2 b_0 \omega_1^2}{8 \omega_2} - \frac{1}{2} e^{i \vartheta 2[t]} a2'[t] - \frac{1}{2} i e^{i \vartheta 2[t]} a2[t] \vartheta 2[t]$$

Autonomous equations and separations of the real and imaginary parts

```
eqa[1] = ComplexExpand[Expand[eq[1] e^(-i ϑ 1[t])]]
```

$$\frac{1}{4} \mu a1[t] - \frac{1}{4} \mu a1[t] a2[t] \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{1}{6} \nu a1[t] a2[t] \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{3}{16} a1[t]^3 b_1 \omega_1^2 + \frac{1}{4} a1[t] a2[t] \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 \omega_2 - \frac{a1'[t]}{2} + i \left(\frac{1}{4} \mu a1[t] a2[t] \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 + \frac{1}{6} \nu a1[t] a2[t] \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{\mu^2 a1[t]}{16 \omega_1} + \frac{3 c a1[t]^3}{16 \omega_1} - \frac{1}{12} a1[t]^3 b_0^2 \omega_1 - \frac{67}{120} a1[t] a2[t]^2 b_0^2 \omega_1 + \frac{1}{4} a1[t] a2[t] \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 \omega_2 - \frac{1}{2} a1[t] \vartheta 1[t] \right)$$

```
eqa[2] = ComplexExpand[Expand[eq[2] e^(-i ϑ 2[t])]]
```

$$-\frac{3}{64} \mu a1[t]^2 \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{1}{64} \sigma a1[t]^2 \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{3}{4} a2[t]^3 b_2 \omega_1^2 + \frac{a1[t]^2 \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 \omega_1^2}{8 \omega_2} - \frac{a2'[t]}{2} + i \left(\frac{1}{2} \sigma a2[t] + \frac{1}{64} \sigma a1[t]^2 \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 - \frac{3}{64} \mu a1[t]^2 \mathrm{Sin}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 + \frac{3 c a2[t]^3}{32 \omega_1} - \frac{61}{240} a1[t]^2 a2[t] b_0^2 \omega_1 - \frac{3}{10} a2[t]^3 b_0^2 \omega_1 - \frac{a1[t]^2 \mathrm{Cos}[2 \vartheta 1[t] - \vartheta 2[t]] b_0 \omega_1^2}{8 \omega_2} - \frac{1}{2} a2[t] \vartheta 2[t] \right)$$

Definition of φ

```
phiRule = {2 ϑ 1[t] - ϑ 2[t] -> φ[t]}
```

$\{2 \vartheta 1[t] - \vartheta 2[t] \to \varphi[t]\}$

```
phiRuleD = {ϑ 2'[t] -> 2 ϑ 1'[t] - φ'[t]}
```

$\{\vartheta 2[t] \to 2 \vartheta 1[t] - \varphi'[t]\}$

Polar Amplitude Modulation Equations

```
amep[1] = Collect[(ComplexExpand[2 eqa[1]] /. ⅈ → 0), {a1[t], a2[t], a1[t] a2[t]}]
```

$$\frac{1}{2}\mu\, a1[t] - \frac{3}{8} a1[t]^3 b_1 \omega_1^2 + a1[t]\, a2[t] \left(-\frac{1}{2}\mu \cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 - \frac{1}{3}\nu \cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 + \frac{1}{2}\sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 \omega_2\right) - a1'[t]$$

```
amep[2] = Collect[(ComplexExpand[2 eqa[2]] /. ⅈ → 0), {a1[t], a2[t], a1[t] a2[t]}]
```

$$-\frac{3}{2} a2[t]^3 b_2 \omega_1^2 + a1[t]^2 \left(-\frac{3}{32}\mu \cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 - \frac{1}{32}\sigma \sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 + \frac{\sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 \omega_1^2}{4\omega_2}\right) - a2'[t]$$

```
amep[3] = Collect[(ComplexExpand[-ⅈ 2 eqa[1]] /. ⅈ → 0), {a1[t], a2[t], a1[t] a2[t]}]
```

$$a1[t]^3 \left(\frac{3c}{8\omega_1} - \frac{1}{6} b_0^2 \omega_1\right) + a1[t]\, a2[t] \left(\frac{1}{2}\mu \sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 + \frac{1}{3}\nu \sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 + \frac{1}{2}\cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 \omega_2\right) + a1[t] \left(-\frac{\mu^2}{8\omega_1} - \frac{67}{60} a2[t]^2 b_0^2 \omega_1 - \vartheta 1[t]\right)$$

```
amep[4] = Collect[(ComplexExpand[-ⅈ 2 eqa[2]] /. ⅈ → 0), {a1[t], a2[t], a1[t] a2[t]}]
```

$$a2[t]^3 \left(\frac{3c}{16\omega_1} - \frac{3}{5} b_0^2 \omega_1\right) + a1[t]^2 \left(\frac{1}{32}\sigma \cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 - \frac{3}{32}\mu \sin[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 - \frac{61}{120} a2[t]\, b_0^2 \omega_1 - \frac{\cos[2\vartheta 1[t] - \vartheta 2[t]]\, b_0 \omega_1^2}{4\omega_2}\right) + a2[t]\, (\sigma - \vartheta 2[t])$$

Reduced Amplitude Modulation Equations

```
rame[1] = Collect[amep[1] /. phiRule, {a1[t], a2[t], a1[t] a2[t]}]
```

$$\frac{1}{2}\mu\, a1[t] - \frac{3}{8} a1[t]^3 b_1 \omega_1^2 + a1[t]\, a2[t] \left(-\frac{1}{2}\mu \cos[\varphi[t]]\, b_0 - \frac{1}{3}\nu \cos[\varphi[t]]\, b_0 + \frac{1}{2}\sin[\varphi[t]]\, b_0 \omega_2\right) - a1'[t]$$

```
rame[2] = Collect[amep[2] /. phiRule, {a1[t], a2[t], a1[t] a2[t]}]
```

$$-\frac{3}{2} a2[t]^3 b_2 \omega_1^2 + a1[t]^2 \left(-\frac{3}{32}\mu \cos[\varphi[t]]\, b_0 - \frac{1}{32}\sigma \sin[\varphi[t]]\, b_0 + \frac{\sin[\varphi[t]]\, b_0 \omega_1^2}{4\omega_2}\right) - a2'[t]$$

```
rame[3] = Collect[Expand[(2 a2[t] amep[3] - a1[t] amep[4]) /. phiRule /. phiRuleD],
  {a1[t], a2[t], a1[t]^3, a2[t]^3, a1[t]^2 a2[t], a1[t] a2[t]^2,
   a1[t]^3 a2[t], a1[t] a2[t]^3, b0, Cos[φ[t]], Sin[φ[t]]}]
```

$$\mathrm{a1}[t]\,\mathrm{a2}[t]^3\left(-\frac{3\,c}{16\,\omega_1}-\frac{49}{30}\,b_0^2\,\omega_1\right)+\mathrm{a1}[t]^3\,\mathrm{a2}[t]\left(\frac{3\,c}{4\,\omega_1}+\frac{7}{40}\,b_0^2\,\omega_1\right)+\mathrm{a1}[t]^3\,b_0\left(\frac{3}{32}\,\mu\,\mathrm{Sin}[\varphi[t]]+\mathrm{Cos}[\varphi[t]]\left(-\frac{\sigma}{32}+\frac{\omega_1^2}{4\,\omega_2}\right)\right)+\mathrm{a1}[t]\,\mathrm{a2}[t]^2\,b_0\left(\left(\mu+\frac{2\,\nu}{3}\right)\mathrm{Sin}[\varphi[t]]+\mathrm{Cos}[\varphi[t]]\,\omega_2\right)+\mathrm{a1}[t]\,\mathrm{a2}[t]\left(-\sigma-\frac{\mu^2}{4\,\omega_1}-\varphi'[t]\right)$$

```
fixRule = {a1[t] -> a1, a2[t] -> a2, φ[t] -> φ}
```

$$\{\mathrm{a1}[t]\to\mathrm{a1},\ \mathrm{a2}[t]\to\mathrm{a2},\ \varphi[t]\to\varphi\}$$

Equations to find Fixed Points

```
fix[1] = Collect[Expand[Simplify[rame[1] /. {a1'[t] → 0, a2'[t] → 0, φ'[t] → 0}]],
   {a1[t], a2[t], a1[t] a2[t], a1[t]^3, a2[t]^3, a1[t]^2 a2[t], a1[t] a2[t]^2}] /. fixRule
```

$$\frac{\mathrm{a1}\,\mu}{2}-\frac{3}{8}\,\mathrm{a1}^3\,b_1\,\omega_1^2+\mathrm{a1}\,\mathrm{a2}\left(-\frac{1}{2}\,\mu\,\mathrm{Cos}[\varphi]\,b_0-\frac{1}{3}\,\nu\,\mathrm{Cos}[\varphi]\,b_0+\frac{1}{2}\,\mathrm{Sin}[\varphi]\,b_0\,\omega_2\right)$$

```
fix[2] =
 Collect[Expand[Simplify[rame[2] /. {a1'[t] → 0, a2'[t] → 0, φ'[t] → 0}]], {a1[t], a2[t],
    a1[t]^3, a2[t]^3, a1[t]^2 a2[t], a1[t] a2[t]^2, b0, Cos[φ[t]], Sin[φ[t]]}] /. fixRule
```

$$-\frac{3}{2}\,\mathrm{a2}^3\,b_2\,\omega_1^2+\mathrm{a1}^2\,b_0\left(-\frac{3}{32}\,\mu\,\mathrm{Cos}[\varphi]+\mathrm{Sin}[\varphi]\left(-\frac{\sigma}{32}+\frac{\omega_1^2}{4\,\omega_2}\right)\right)$$

```
fix[3] = Collect[Expand[Simplify[rame[3] /. {a1'[t] → 0, a2'[t] → 0, φ'[t] → 0}]],
   {a1[t], a2[t], a1[t]^3, a2[t]^3, a1[t]^2 a2[t], a1[t] a2[t]^2,
    a1[t]^3 a2[t], a1[t] a2[t]^3, b0, Cos[φ[t]], Sin[φ[t]]}] /. fixRule
```

$$\mathrm{a1}\,\mathrm{a2}\left(-\sigma-\frac{\mu^2}{4\,\omega_1}\right)+\mathrm{a1}\,\mathrm{a2}^3\left(-\frac{3\,c}{16\,\omega_1}-\frac{49}{30}\,b_0^2\,\omega_1\right)+\mathrm{a1}^3\,\mathrm{a2}\left(\frac{3\,c}{4\,\omega_1}+\frac{7}{40}\,b_0^2\,\omega_1\right)+\mathrm{a1}^3\,b_0\left(\frac{3}{32}\,\mu\,\mathrm{Sin}[\varphi]+\mathrm{Cos}[\varphi]\left(-\frac{\sigma}{32}+\frac{\omega_1^2}{4\,\omega_2}\right)\right)+\mathrm{a1}\,\mathrm{a2}^2\,b_0\left(\left(\mu+\frac{2\,\nu}{3}\right)\mathrm{Sin}[\varphi]+\mathrm{Cos}[\varphi]\,\omega_2\right)$$

Equilibrium paths in the case $a_2 = 0$

```
ep[1] = Simplify[Solve[(fix[1] /. a2 → 0) == 0, a1], ω 1> 0]
```

$$\left\{\{\mathrm{a1}\to 0\},\ \left\{\mathrm{a1}\to -\frac{2\sqrt{\mu}}{\sqrt{3}\,\sqrt{b_1}\,\omega_1}\right\},\ \left\{\mathrm{a1}\to \frac{2\sqrt{\mu}}{\sqrt{3}\,\sqrt{b_1}\,\omega_1}\right\}\right\}$$

Reconstitution of the solution

```
qr₁[t_] = ComplexExpand[
  (A₁[T₁, T₂] ⅇ^(ⅈ ω₁ t) + Ā₁[T₁, T₂] ⅇ^(-ⅈ ω₁ t) + solq11) /. traspRule /. polRule /. {T₀ -> t}]
```

$$\text{a1}[t]\,\text{Cos}[t\,\omega_1+\vartheta 1[t]]+\frac{1}{2}\,\text{a1}[t]^2\,b_0+\frac{1}{6}\,\text{a1}[t]^2\,\text{Cos}[2\,t\,\omega_1+2\,\vartheta 1[t]]\,b_0-\frac{\text{a1}[t]\,\text{a2}[t]\,\text{Cos}[3\,t\,\omega_1+\vartheta 1[t]+\vartheta 2[t]]\,b_0\,\omega_2}{8\,\omega_1}+\frac{\text{a2}[t]^2\,b_0\,\omega_2^2}{2\,\omega_1^2}+\frac{\text{a2}[t]^2\,\text{Cos}[4\,t\,\omega_1+2\,\vartheta 2[t]]\,b_0\,\omega_2^2}{30\,\omega_1^2}$$

```
qr₂[t_] = ComplexExpand[
  (A₂[T₁, T₂] ⅇ^(ⅈ ω₂ t) + Ā₂[T₁, T₂] ⅇ^(-ⅈ ω₂ t) + solq12) /. traspRule /. polRule /. {T₀ -> t}]
```

$$\text{a2}[t]\,\text{Cos}[t\,\omega_2+\vartheta 2[t]]-\frac{1}{2}\,\text{a2}[t]^2\,b_0-\frac{1}{6}\,\text{a2}[t]^2\,\text{Cos}[2\,t\,\omega_2+2\,\vartheta 2[t]]\,b_0-\frac{4\,\nu\,\text{a1}[t]\,\text{Sin}\left[\frac{t\,\omega_2}{2}+\vartheta 1[t]\right]\omega_1}{3\,\omega_2^2}-\frac{\text{a1}[t]^2\,b_0\,\omega_1^2}{2\,\omega_2^2}+\frac{4\,\text{a1}[t]\,\text{a2}[t]\,\text{Cos}\left[\frac{t\,\omega_2}{2}-\vartheta 1[t]+\vartheta 2[t]\right]b_0\,\omega_1}{3\,\omega_2}+\frac{4\,\text{a1}[t]\,\text{a2}[t]\,\text{Cos}\left[\frac{3\,t\,\omega_2}{2}+\vartheta 1[t]+\vartheta 2[t]\right]b_0\,\omega_1}{5\,\omega_2}$$

Numerical integrations

Numerical values

```
{ω₁ = 1, ω₂ = 2, b₀ = 1/2, b₁ = 1, b₂ = 1, c = 1, μ = 0.05, ν = 0.05, σ = 0, ω̄₂ = 2 ω₁ + σ}
```

$$\left\{1,\,2,\,\frac{1}{2},\,1,\,1,\,1,\,0.05,\,0.05,\,0,\,2\right\}$$

Time of integration

```
ti = 200;
```

Numerical Intergations of the RAME

```
solrame[1] = NDSolve[{rame[1] == 0, rame[2] == 0, rame[3] == 0,
   a1[0] == 0.1, a2[0] == 0.1, φ[0] == 0.1}, {a1[t], a2[t], φ[t]}, {t, 0, ti}]
```

```
{{a1[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  a2[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  φ[t] → InterpolatingFunction[{{0., 200.}}, <>][t]}}
```

```
GraphicsArray[{Plot[{a1[t] /. solrame[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.8, 0.8}}, Frame → True, FrameLabel -> {"t", "a1(t)"}],
  Plot[{a2[t] /. solrame[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.6, 0.4}}, Frame → True, FrameLabel -> {"t", "a2(t)"}],
  Plot[{φ[t] /. solrame[1]}, {t, 0, ti}, PlotStyle → Thick, Frame → True,
    AxesOrigin -> {0, 0}, FrameLabel -> {"t", "φ(t)"}]}]
```



Numerical Intergations of the AME

```
solramep[1] = NDSolve[{amep[1] == 0, amep[2] == 0, amep[3] == 0, amep[4] == 0, a1[0] == 0.1,
    a2[0] == 0.1, ϑ1[0] == 0.1, ϑ2[0] == 0.1}, {a1[t], a2[t], ϑ1[t], ϑ2[t]}, {t, 0, ti}]
```

```
{{a1[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  a2[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  ϑ1[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  ϑ2[t] → InterpolatingFunction[{{0., 200.}}, <>][t]}}
```

```
GraphicsArray[{Plot[{a1[t] /. solramep[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.8, 0.8}}, Frame → True, FrameLabel -> {"t", "a1(t)"}],
  Plot[{a2[t] /. solramep[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.6, 0.4}}, Frame → True, FrameLabel -> {"t", "a2(t)"}],
  Plot[{ϑ1[t] /. solramep[1]}, {t, 0, ti}, PlotStyle → Thick,
    Frame → True, AxesOrigin -> {0, 0}, FrameLabel -> {"t", "ϑ1(t)"}],
  Plot[{ϑ2[t] /. solramep[1]}, {t, 0, ti}, PlotStyle → Thick,
    Frame → True, AxesOrigin -> {0, 0}, FrameLabel -> {"t", "ϑ2(t)"}]}]
```



Graphics of the reconstituted solution

```
GraphicsArray[{Plot[qr1[t] /. solramep[1], {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.8, 0.8}}, Frame → True, FrameLabel -> {"t", "q1(t)"}],
  Plot[qr2[t] /. solramep[1], {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.6, 0.4}}, Frame → True, FrameLabel -> {"t", "q2(t)"}]}]
```



Numerical Intergations of the original equations

```
solorig[1] = NDSolve[Join[EOM, {q1[0] == 0.1, q2[0] == 0.1, q1'[0] == 0.1, q2'[0] == 0.1}],
  {q1[t], q2[t]}, {t, 0, ti}, MaxSteps -> 1 000 000]

{{q1[t] → InterpolatingFunction[{{0., 200.}}, <>][t],
  q2[t] → InterpolatingFunction[{{0., 200.}}, <>][t]}}

GraphicsArray[{Plot[{q1[t] /. solorig[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.8, 0.8}}, Frame → True, FrameLabel -> {"t", "q1(t)"}],
  Plot[{q2[t] /. solorig[1]}, {t, 0, ti}, PlotStyle → Thick,
    PlotRange -> {Automatic, {-0.6, 0.4}}, Frame → True, FrameLabel -> {"t", "q2(t)"}]}]
```